what-when-how

In Depth Tutorials and Information

LCD INTERFACING

This chapter explores some real-world applications of the 8051. We explain how to interface the 8051 to devices such as an LCD and a keyboard. In Section 12.1, we show LCD interfacing with the 8051. In Section 12.2, keyboard interfacing with the 8051 is shown. We use C and Assembly for both sections.

SECTION 12.1: LCD INTERFACING

This section describes the operation modes of LCDs, then describes how to program and interface an LCD to an 8051 using Assembly and C.

LCD operation

In recent years the LCD is finding widespread use replacing LEDs (seven-segment LEDs or other multisegment LEDs). This is due to the following reasons:

1. The declining prices of LCDs.
    1. The ability to display numbers, characters, and graphics. This is in contrast to LEDs, which are limited to numbers and a few characters.
    2. Incorporation of a refreshing controller into the LCD, thereby relieving the CPU of the task of refreshing the LCD. In contrast, the LED must be refreshed by the CPU (or in some other way) to keep displaying the data.
2. Ease of programming for characters and graphics.

LCD pin descriptions

The LCD discussed in this section has 14 pins. The function of each pin is given in Table 12-1. Figure 12-1 shows the pin positions for various LCDs.

Table 12-1: Pin Descriptions for LCD

Pin	Symbol	I/O	Description
1	V_{SS}	--	Ground
2	V_{CC}	--	+5V power supply
3	V_{EE}	--	Power supply to control contrast
4	RS	I	RS = 0 to select command register, RS = 1 to select data register
5	R/W	I	R/W = 0 for write, R/W = 1 for read
6	E	I/O	Enable
7	DB0	I/O	The 8-bit data bus
8	DB1	I/O	The 8-bit data bus
9	DB2	I/O	The 8-bit data bus
10	DB3	I/O	The 8-bit data bus
11	DB4	I/O	The 8-bit data bus
12	DB5	I/O	The 8-bit data bus
13	DB6	I/O	The 8-bit data bus
14	DB7	I/O	The 8-bit data bus

V_{CC}, V_{SS}, and V_{EE}

While V_{cc} and V_{ss} provide +5V and ground, respectively, V_{EE} is used for controlling LCD contrast.

RS, register select

There are two very important registers inside the LCD. The RS pin is used for their selection as follows. If RS = 0, the instruction command code register is selected, allowing the user to send a command such as clear display, cursor at home, etc. If RS = 1 the data register is selected, allowing the user to send data to be displayed on the LCD.

R/W, read/write

R/W input allows the user to write information to the LCD or read information from it. R/W = 1 when reading; R/W = 0 when writing.

E, enable

The enable pin is used by the LCD to latch information presented to its data pins.

When data is supplied to data pins, a high-to-low pulse must be applied to this pin in order for the LCD to latch in the data present at the data pins. This pulse must be a minimum of 450 ns wide.

DO-D7

The 8-bit data pins, DO – D7, are used to send information to the LCD or read the contents of the LCD's internal registers.

To display letters and numbers, we send ASCII codes for the letters A – Z, a – z, and numbers 0 – 9 to these pins while making RS = 1.

There are also instruc-

Table 12-2: LCD Command Codes

Code (Hex)	Command to LCD Instruction Register
1	Clear display screen
2	Return home
4	Decrement cursor (shift cursor to left)
6	Increment cursor (shift cursor to right)
5	Shift display right
7	Shift display left
8	Display off, cursor off
A	Display off, cursor on
C	Display on, cursor off
E	Display on, cursor blinking
F	Display on, cursor blinking
10	Shift cursor position to left
14	Shift cursor position to right
18	Shift the entire display to the left
1C	Shift the entire display to the right
80	Force cursor to beginning of 1st line
C0	Force cursor to beginning of 2nd line
38	2 lines and 5x7 matrix

Note: This table is extracted from Table 12-4.

tion command codes that can be sent to the LCD to clear the display or force the cursor to the home position or blink the cursor. Table 12-2 lists the instruction command codes.

We also use RS = 0 to check the busy flag bit to see if the LCD is ready to receive information. The busy flag is D7 and can be read when R/W = 1 and RS = 0, as follows: if R/W= 1, RS = 0. When D7 = 1 (busy flag = 1), the LCD is busy taking care of internal operations and will not accept any new information. When D7 = 0, the LCD is ready to receive new information. *Note:* It is recommended to check the busy flag before writing any data to the LCD.



Figure 12-1. Pin Positions for Various LCDs from Optrex

Sending commands and data to LCDs with a time delay

To send any of the commands from Table 12-2 to the LCD, make pin RS = 0. For data, make RS = 1. Then send a high-to-low pulse to the E pin to enable the internal latch of the LCD. This is shown in Program 12-1. See Figure 12-2 for LCD connections.

```
;calls a time delay before sending next data/command
; P1.0-P1.7 are connected to LCD data pins D0-D7
; P2.0 is connected to RS pin of LCD
; P2.1 is connected to R/W pin of LCD
; P2.2 is connected to E pin of LCD
        ORG   0H
        MOV   A,#38H          ;init. LCD 2 lines,5x7 matrix
        ACALL COMNWRT         ;call command subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#0EH          ;display on, cursor on
        ACALL COMNWRT         ;call command subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#01           ;clear LCD
        ACALL COMNWRT         ;call command subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#06H          ;shift cursor right
        ACALL COMNWRT         ;call command subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#84H          ;cursor at line 1,pos. 4
        ACALL COMNWRT         ;call command subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#'N'          ;display letter N
        ACALL DATAWRT         ;call display subroutine
        ACALL DELAY           ;give LCD some time
        MOV   A,#'O'          ;display letter O
        ACALL DATAWRT         ;call display subroutine
AGAIN:  SJMP  AGAIN           ;stay here
COMNWRT:                      ;send command to LCD
        MOV   P1,A            ;copy reg A to port1
        CLR   P2.0            ;RS=0 for command
        CLR   P2.1            ;R/W=0 for write
        SETB  P2.2            ;E=1 for high pulse
        ACALL DELAY           ;give LCD some time
        CLR   P2.2            ;E=0 for H-to-L pulse
        RET
DATAWRT:                      ;write data to LCD
        MOV   P1,A            ;copy reg A to port1
        SETB  P2.0            ;RS=1 for data
        CLR   P2.1            ;R/W=0 for write
        SETB  P2.2            ;E=1 for high pulse
        ACALL DELAY           ;give LCD some time
        CLR   P2.2            ;E=0 for H-to-L pulse
        RET
```

Program 12-1: Communicating with LCD using a delay

```
DELAY:  MOV   R3,#50          ;50 or higher for fast CPUs
HERE2:  MOV   R4,#255         ;R4=255
HERE:   DJNZ  R4,HERE         ;stay until R4 becomes 0
        DJNZ  R3,HERE2
        RET
        END
```

Sending code or data to the LCD with checking busy flag



The above code showed how to send commands to the LCD without checking the busy flag. Notice that we must put a long delay between issuing data or commands to the LCD. However, a much better way is to monitor the busy flag before issuing a command or data to the LCD. This is shown in Program 12-2.

Figure 12-2. LCD Connections

```
;Check busy flag before sending data, command to LCD
;P1=data pin,P2.0=RS,P2.1=R/W,P2.2=E pins
         MOV   A,#38H          ;init. LCD 2 lines,5x7 matrix
         ACALL COMMAND         ;issue command
         MOV   A,#0EH          ;LCD on, cursor on
         ACALL COMMAND         ;issue command
         MOV   A,#01H          ;clear LCD command
         ACALL COMMAND         ;issue command
         MOV   A,#06H          ;shift cursor right
         ACALL COMMAND         ;issue command
         MOV   A,#86H          ;cursor: line 1, pos. 6
         ACALL COMMAND         ;command subroutine
         MOV   A,#'N'          ;display letter N
         ACALL DATA_DISPLAY
         MOV   A,#'O'          ;display letter O
         ACALL DATA_DISPLAY
HERE:    SJMP  HERE            ;STAY HERE
COMMAND: ACALL READY           ;is LCD ready?
         MOV   P1,A            ;issue command code
         CLR   P2.0            ;RS=0 for command
         CLR   P2.1            ;R/W=0 to write to LCD
         SETB  P2.2            ;E=1 for H-to-L pulse
         CLR   P2.2            ;E=0 ,latch in
         RET
```

Program 12-2: Communicating with LCD using the busy flag

```
DATA_DISPLAY:
          ACALL READY          ;is LCD ready?
          MOV   P1,A           ;issue data
          SETB  P2.0           ;RS=1 for data
          CLR   P2.1           ;R/W=0 to write to LCD
          SETB  P2.2           ;E=1 for H-to-L pulse
          ACALL DELAY          ;give LCD some time
          CLR   P2.2           ;E=0, latch in
          RET
READY:    SETB  P1.7           ;make P1.7 input port
          CLR   P2.0           ;RS=0 access command reg
          SETB  P2.1           ;R/W=1 read command reg
;read command reg and check busy flag
BACK:     CLR   P2.2           ;E=0 for L-to-H pulse
          ACALL DELAY          ;give LCD some time
          SETB  P2.2           ;E=1 L-to-H pulse
          JB    P1.7,BACK      ;stay until busy flag=0
          RET
          END
```



t_D = Data output delay time
t_{AS} = Setup time prior to E (going high) for both RS and R/W = 140 ns (minimum)
t_{AH} = Hold time after E has come down for both RS and R/W = 10 ns (minimum)

Note: Read requires an L-to-H pulse for the E pin.

Figure 12-3. LCD Timing for Read (L-to-H for E line)

Notice in the above program that the busy flag is D7 of the command register. To read the command register we make R/W = 1 and RS = 0, and a L-to-H pulse for the E pin will provide us the command register. After reading the command register, if bit D7 (the busy flag) is high, the LCD is busy and no information (command or data) should be issued to it. Only when D7 = 0 can we send data or commands to the LCD. Notice in this method that no time delays are used since we are checking the busy flag before issuing commands or data to the LCD. Contrast the Read and Write timing for the LCD in Figures 12-3 and 12-4. Note that the E line is negative-edge triggered for the write while it is positive-edge triggered for the read.



t_{PWH} = Enable pulse width = 450 ns (minimum)
t_{DSW} = Data setup time = 195 ns (minimum)
t_H = Data hold time = 10 ns (minimum)
t_{AS} = Setup time prior to E (going high) for both RS and R/W = 140 ns (minimum)
t_{AH} = Hold time after E has come down for both RS and R/W = 10 ns (minimum)

Figure 12-4. LCD Timing for Write (H-to-L for E line)

LCD data sheet

In the LCD, one can put data at any location. The following shows address locations and how they are accessed.

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	1	A	A	A	A	A	A	A

where AAAAAAA = 0000000 to 0100111 for line 1 and AAAAAAA – 1000000 to 1100111 for line 2. See Table 12-3.

Table 12-3: LCD Addressing

	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
Line 1 (min)	1	0	0	0	0	0	0	0
Line 1 (max)	1	0	1	0	0	1	1	1
Line 2 (min)	1	1	0	0	0	0	0	0
Line 2 (max)	1	1	1	0	0	1	1	1

The upper address range can go as high as 0100111 for the 40-charac-ter-wide LCD, while for the 20-character-wide LCD it goes up to 010011 (19 decimal = 10011 binary). Notice that the upper range 0100111 (binary) = 39 decimal, which corresponds to locations 0 to 39 for the LCDs of 40×2 size.

16 x 2 LCD	80	81	82	83	84	85	86 through 8F
	C0	C1	C2	C3	C4	C5	C6 through CF
20 x 1 LCD	80	81	82	83	through 93		
20 x 2 LCD	80	81	82	83	through 93		
	C0	C1	C2	C3	through D3		
20 x 4 LCD	80	81	82	83	through 93		
	C0	C1	C2	C3	through D3		
	94	95	96	97	through A7		
	D4	D5	D6	D7	through E7		
40 x 2 LCD	80	81	82	83	through A7		
	C0	C1	C2	C3	through E7		

Note: All data is in hex.

Figure 12-5. Cursor Addresses for Some LCDs

From the above discussion we can get the addresses of cursor positions for various sizes of LCDs. See Figure 12-5 for the cursor addresses for common types of LCDs. Note that all the addresses are in hex. Table 12-4 provides a detailed list of LCD commands and instructions. Table 12-2 is extracted from this table.

Optrex is one of the largest manufacturer of LCDs. You can obtain datasheets from their Web site, www.optrex.com.

The LCDs can be purchased from the following Web sites:

www. digikey. com
www.jameco.com
www.elexp.com

Table 12-4: List of LCD Instructions

Instruction	RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0	Description	Execution Time (Max)
Clear Display	0	0	0	0	0	0	0	0	0	1	Clears entire display and sets DD RAM address 0 in address counter	1.64 ms
Return Home	0	0	0	0	0	0	0	0	1	-	Sets DD RAM address 0 as address counter. Also returns display being shifted to original position. DD RAM contents remain unchanged.	1.64 ms
Entry Mode Set	0	0	0	0	0	0	0	1	I/D	S	Sets cursor move direction and specifies shift of display. These operations are performed during data write and read.	40 µs
Display On/ Off Control	0	0	0	0	0	0	1	D	C	B	Sets On/Off of entire display (D), cursor On/Off (C), and blink of cursor position character (B).	40 µs
Cursor or Display Shift	0	0	0	0	0	1	S/C	R/L	-	-	Moves cursor and shifts display without changing DD RAM contents.	40 µs
Function Set	0	0	0	0	1	DL	N	F	-	-	Sets interface data length (DL), number of display lines (L), and character font (F).	40 µs
Set CG RAM Address	0	0	0	1	AGC						Sets CG RAM address. CG RAM data is sent and received after this setting.	40 µs
Set DD RAM Address	0	0	1	ADD							Sets DD RAM address. DD RAM data is sent and received after this setting.	40 µs
Read Busy Flag & Address	0	1	BF	AC							Reads Busy flag (BF) indicating internal operation is being performed and reads address counter contents.	40 µs
Write Data CG or DD RAM	1	0	Write Data								Writes data into DD or CG RAM.	40 µs
Read Data CG or DD RAM	1	1	Read Data								Reads data from DD or CG RAM.	40 µs

Notes:

1. Execution times are maximum times when fcp or fosc is 250 kHz.
2. Execution time changes when frequency changes. Ex: When fcp or fosc is 270 kHz: 40 µs × 250 / 270 = 37 µs.
3. Abbreviations:

DD RAM	Display data RAM	
CG RAM	Character generator RAM	
ACC	CG RAM address	
ADD	DD RAM address, corresponds to cursor address	
AC	Address counter used for both DD and CG RAM addresses.	
I/D = 1	Increment	I/D = 0 Decrement
S = 1	Accompanies display shift	
S/C = 1	Display shift;	S/C = 0 Cursor move
R/L = 1	Shift to the right;	R/L = 0 Shift to the left
DL = 1	8 bits, DL = 0: 4 bits	
N = 1	1line, N = 0 : 1 line	
F = 1	5 x 10 dots, F = 0 : 5 x 7 dots	
BF = 1	Internal operation;	BF = 0 Can accept instruction

Sending information to LCD using MOVC instruction

The Program 12-3 shows how to use the MOVC instruction to send data and commands to an LCD. For an 8051 C version'of LCD programming see Examples 12-1 and 12-2.

```
;calls a time delay before sending next data/command
; P1.0-P1.7=D0-D7, P2.0=RS, P2.1=R/W, P2.2=E pins
          ORG   0
          MOV   DPTR,#MYCOM
C1:       CLR   A
          MOVC  A,@A+DPTR
          ACALL COMNWRT         ;call command subroutine
          ACALL DELAY           ;give LCD some time
          JZ    SEND_DAT
          INC   DPTR
          SJMP  C1
SEND_DAT: MOV   DPTR,#MYDATA
D1:       CLR   A
          MOVC  A,@A+DPTR
          ACALL DATAWRT         ;call command subroutine
          ACALL DELAY           ;give LCD some time
          INC   DPTR
          JZ    AGAIN
          SJMP  D1
AGAIN:    SJMP  AGAIN           ;stay here
COMNWRT:                        ;send command to LCD
          MOV   P1,A            ;SEND COMND to P1
          CLR   P2.0            ;RS=0 for command
          CLR   P2.1            ;R/W=0 for write
          SETB  P2.2            ;E=1 for high pulse
          ACALL DELAY           ;give LCD some time
          CLR   P2.2            ;E=0 for H-to-L
          RET
DATAWRT:
          MOV   P1,A            ;SEND DATA to P1
          SETB  P2.0            ;RS=1 for data
          CLR   P2.1            ;R/W=0 for write
          SETB  P2.2            ;E=1 for high pulse
          ACALL DELAY           ;give LCD some time
          CLR   P2.2            ;E=0 for H-to-L pulse
          RET
DELAY:    MOV   R3,#250         ;LONG DELAY FOR fast CPUs
HERE2:    MOV   R4,#255         ;
HERE:     DJNZ  R4,HERE         ;
          DJNZ  R3,HERE2
          RET
          ORG   300H
MYCOM:    DB    38H,0EH,01,06,84H,0     ;commands and null
MYDATA:   DB    "HELLO",0               ;data and null
          END
```

Program 12-3: Sending information to LCD with MOVC instruction.

You can also see introduction in
https://www.8051projects.net/lcd-interfacing/introduction.php